
Dependent types

I In Coq, types may be parameterized by values.

I Such types are called dependent.



Example of dependent types

I Arrays of size n, perfect binary trees of depth p, ...

I Logical formulas!
I Universally quantified theorems are functions
I Application is instanciation
I Propositions are types, proofs are elements

I Partial functions handled via (preconditions):
I pred safe : forall x:nat, x<>0 -> nat

Example : a total predecessor

Definition pred_safe (n:nat) : n<>O -> nat :=

match n with

| O => fun Hn => False_rect _ (Hn (eq_refl O))

| S n => fun _ => n

end.

(* or *)

Definition pred_safe : forall n, n<>O -> nat.

Proof.

intros n Hn. destruct n.

destruct Hn; reflexivity.

apply n.

Qed.

Example : bounded numbers and arrays

Inductive bnat (n : nat) : Type :=

cb : forall m, m < n -> bnat n.

Inductive array (n : nat) : Type :=

ca : forall l : list Z, length l = n -> array n.

(* or *)

Inductive vect : nat -> Type :=

| vnil : vect 0

| vcons : forall n, Z -> vect n -> vect (S n).

We can build a total nth function:

Definition vect_nth : forall n, vect n -> bnat n -> Z.

Proof. ... Defined.

Example : perfect binary tree

Fixpoint ptree (n:nat) : Type :=

match n with

| O => Z

| S n => (ptree n * ptree n)%type

end.

Check ((1,2),(3,4)) : ptree 2.

Fixpoint sum_ptree n : ptree n -> Z :=

match n with

| O => fun t => t

| S n =>

fun t => let (g,d):=t in sum_ptree n g + sum_ptree n d

end.

Compute (sum_ptree 2 ((1,2),(3,4))).



A generic notion of type with restriction

I bnat and array are quite similar:
numbers, or lists, such that some property hold.

I Coq’s generic way to build types with restriction:

{ x : A | P x }

I For instance:

Definition bnat n := { m | m < n }.

Definition array n := { l : list Z | length l = n }.

A generic notion of type with restriction

I Behind the nice { | } notation, the sig type:

Inductive sig (A : Type) (P : A -> Prop) : Type :=

exist : forall x : A, P x -> sig P

I To access the element, or the proof of the property:

I proj1 sig, proj2 sig
I or directly let (x,p) := ... in ...
I or in proof mode via the tactics case, destruct, ...

I To build a sig interactively: the exists tactic.

A example: bounded successor

I As a function:

Definition bsucc n : bnat n -> bnat (S n) :=

fun m => let (x,p):= m in exist _ (S x) (lt_n_S _ _ p).

I Via tactics:

Definition bsucc n : bnat n -> bnat (S n).

Proof.

intros m. destruct m as [x p]. exists (S x).

auto with arith.

Defined.

I Via the Program framework :

Program Definition bsucc n : bnat n -> bnat (S n) :=

fun m => S m.

Next Obligation.

destruct m. simpl. auto with arith.

Qed.

General shape of a rich specification

I With sig, we can hence express also post-conditions:

forall x, P x -> { y | Q x y }

I Adapt to your needs: multiple arguments or outputs (y can be
a tuple) or pre or post (Q can be a conjonction).

I Apart with Program, sig is rarely used for pre-conditions.



The special case of boolean output

I We could handle boolean outputs via sig:

Definition rich_beq_nat :

forall n m : nat, { b : bool | b = true <-> n=m }.

I More convenient: sumbool, a type with two alternatives and
annotations for characterizing them.

Definition eq_nat_dec :

forall n m : nat, { n=m }+{ n<>m }.

The special case of boolean output

I Behind the { }+{ } notation, the sumbool type:

Inductive sumbool (A B : Prop) : Type :=

| left : A -> {A}+{B}

| right : B -> {A}+{B}

I To analyse a sumbool construction:

I directly via if ... then ... else ...
I or bool of sumbool
I or in proof mode via the tactics case, destruct, ...

I To build a sumbool interactively: the left and right tactics.

Decidability result

I Many Coq functions are currently formulated this way:
eq nat dec, Z eq dec, le lt dec, ...
(see SearchAbout sumbool).

I For instance:

Definition le lt dec n m : { n <= m }+{ m < n }.

Proof.

induction n.

left. auto with arith.

destruct m.

right. auto with arith.

destruct (IHn m); [left | right]; auto with arith.

Defined.

I For equality, see tactic decide equality.

Why program with logical annotations ?

I To handle partial functions, instead of dummy values at
undefined spots or option types

I To satisfy precisely an interface (see exercise on sets)

I To have all-in-one objects (handy for destruct).

I To have the right justifications when doing general recursion

Additional remarks:

I Computations in Coq may then be tricky and/or slower
and/or memory hungry.

I Pure & efficient Ocaml/Haskell code can be obtained by
extraction.

I Definitions by tactics are dreadful, Program helps but is still
quite young.

I Instead of destructing rich objects, other technics can also be
convenient (iff, reflect).

Why specific constructs like sig and sumbool ?

I { x | P x } is a clone of exists x, P x.
Both regroup a witness and a justification.

I Similarly, { A }+{ B } is a clone of A \/ B .

In fact, sig/sumbool live in a different world than ex/or.

The two worlds of Coq

In Coq, two separate worlds (technically, we speak of sorts):

I The “logical” world

I a proof : a statement : Prop
I or introl I : True\/False : Prop

I The “informative” world (everything else).

I a program : a type : Type
I O : nat : Type
I pred : nat->nat : Type

The two worlds of Coq

Usually we program in Type and make proofs in Prop. But that’s
just a convention. We can build functions by tactics, or
reciprocally “program” a proof:

Definition or_sym A B : A\/B -> B\/A :=

fun h => match h with

| or_introl a => or_intror _ a

| or_intror b => or_introl _ b

end.

The similarity between proofs and programs, between statements
and types is called the Curry-Howard isomorphism.



The two worlds of Coq

In Coq, a rigid separation between Prop and Type:

Logical parts should not interfere with computations in Type.

Definition nat_of_or A B : A\/B -> nat :=

fun h => match h with

| or_introl _ => 0

| or_intror _ => 1

end.

Error: ... proofs can be eliminated only to build proofs.

Idea: proofs are there only as guarantee, we’re interested only in
their existence, we consider them as having no computational
content.



Extraction
Coq’s strict separation between Prop and Type is the fondation of
the extraction mechanism: roughly, logical parts are removed,
pruned programs still compute the same outputs.

Coq < Recursive Extraction le_lt_dec.

type nat = O | S of nat

type sumbool = Left | Right

(** val le_lt_dec : nat -> nat -> sumbool **)

let rec le_lt_dec n m =

match n with

| O -> Left

| S n0 -> (match m with

| O -> Right

| S m0 -> le_lt_dec n0 m0)


